Rules for integrands of the form $(c + d x)^m \left(a + b \left(F^{g (e+f x)}\right)^n\right)^p$

1. $\int (c + d x)^m \left(b F^{g (e+f x)}\right)^n \, dx$

- If the control variable `$UseGamma` is `True`, antiderivatives of expressions of the form $(d + e x)^m \left(F^{c (a+b x)}\right)^n$ will be much more compactly expressed in terms of the `Gamma` function instead of elementary functions.

```
$UseGamma=False;
```

1: $\int (c + d x)^m \left(b F^{g (e+f x)}\right)^n \, dx$ when $m > 0 \land 2 m \in \mathbb{Z}$

- Derivation: Integration by parts

- Basis: $\left(b F^{g (e+f x)}\right)^n = \partial_x \frac{\left(b F^{g (e+f x)}\right)^n}{f g n \operatorname{Log}[F]}$

- Rule: If $m > 0 \land 2 m \in \mathbb{Z}$, then

$$\int (c + d x)^m \left(b F^{g (e+f x)}\right)^n \, dx \longrightarrow \frac{(c + d x)^m \left(b F^{g (e+f x)}\right)^n}{f g n \operatorname{Log}[F]} - \frac{d m}{f g n \operatorname{Log}[F]} \int (c + d x)^{m-1} \left(b F^{g (e+f x)}\right)^n \, dx$$

- Program code:

```
Int[(c_.+d_.*x_)^m_.*(b_.*F_^(g_.*(e_.+f_.*x_)))^n_.,x_Symbol] :=
  (c+d*x)^m*(b*F^(g*(e+f*x)))^n/(f*g*n*Log[F]) -
  d*m/(f*g*n*Log[F])*Int[(c+d*x)^(m-1)*(b*F^(g*(e+f*x)))^n,x] /;
FreeQ[{F,b,c,d,e,f,g,n},x] && GtQ[m,0] && IntegerQ[2*m] && Not[TrueQ[$UseGamma]]
```

2: $\int (c+d\,x)^m \left(b\,F^{g\,(e+f\,x)}\right)^n\,dx$ when $m < -1 \land 2\,m \in \mathbb{Z}$

- **Derivation: Integration by parts**
- **Basis:** $(c+d\,x)^m == \partial_x \frac{(c+d\,x)^{m+1}}{d\,(m+1)}$
- **Rule: If** $m < -1 \land 2\,m \in \mathbb{Z}$**, then**

$$\int (c+d\,x)^m \left(b\,F^{g\,(e+f\,x)}\right)^n\,dx \longrightarrow \frac{(c+d\,x)^{m+1} \left(b\,F^{g\,(e+f\,x)}\right)^n}{d\,(m+1)} - \frac{f\,g\,n\,\mathrm{Log}[F]}{d\,(m+1)} \int (c+d\,x)^{m+1} \left(b\,F^{g\,(e+f\,x)}\right)^n\,dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_*(b_.*F_^(g_.*(e_.+f_.*x_)))^n_.,x_Symbol] :=
  (c+d*x)^(m+1)*(b*F^(g*(e+f*x)))^n/(d*(m+1)) -
  f*g*n*Log[F]/(d*(m+1))*Int[(c+d*x)^(m+1)*(b*F^(g*(e+f*x)))^n,x] /;
FreeQ[{F,b,c,d,e,f,g,n},x] && LtQ[m,-1] && IntegerQ[2*m] && Not[TrueQ[$UseGamma]]
```

3. $\int (c+d\,x)^m F^{g\,(e+f\,x)}\,dx$

1. $\int (c+d\,x)^m F^{g\,(e+f\,x)}\,dx$ when $m \in \mathbb{Z}$

1: $\int \frac{F^{g\,(e+f\,x)}}{c+d\,x}\,dx$

- **Basis:** $\mathrm{ExpIntegralEi}'[z] == \frac{e^z}{z}$

Rule:

$$\int \frac{F^{g\,(e+f\,x)}}{c+d\,x}\,dx \longrightarrow \frac{1}{d} F^{g\left(e-\frac{c\,f}{d}\right)} \mathrm{ExpIntegralEi}\left[\frac{f\,g\,(c+d\,x)\,\mathrm{Log}[F]}{d}\right]$$

Program code:

```
Int[F_^(g_.*(e_.+f_.*x_))/(c_.+d_.*x_),x_Symbol] :=
  F^(g*(e-c*f/d))/d*ExpIntegralEi[f*g*(c+d*x)*Log[F]/d] /;
FreeQ[{F,c,d,e,f,g},x] && Not[TrueQ[$UseGamma]]
```

2: $\int (c+d\,x)^m\,F^{g\,(e+f\,x)}\,dx$ when $m\in\mathbb{Z}$

Rule: If $m\in\mathbb{Z}$, then

$$\int (c+d\,x)^m\,F^{g\,(e+f\,x)}\,dx\;\longrightarrow\;\frac{(-d)^m\,F^{g\,\left(e-\frac{c\,f}{d}\right)}}{f^{m+1}\,g^{m+1}\,\mathrm{Log}[F]^{m+1}}\,\mathrm{Gamma}\!\left[m+1,\;-\frac{f\,g\,\mathrm{Log}[F]}{d}\,(c+d\,x)\right]$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*F_^(g_.*(e_.+f_.*x_)),x_Symbol] :=
  (-d)^m*F^(g*(e-c*f/d))/(f^(m+1)*g^(m+1)*Log[F]^(m+1))*Gamma[m+1,-f*g*Log[F]/d*(c+d*x)] /;
FreeQ[{F,c,d,e,f,g},x] && IntegerQ[m]
```

2. $\int (c+d\,x)^m\,F^{g\,(e+f\,x)}\,dx$ when $m\notin\mathbb{Z}$

1: $\int \frac{F^{g\,(e+f\,x)}}{\sqrt{c+d\,x}}\,dx$

Derivation: Integration by substitution

- **Basis:** $\frac{F^{g\,(e+f\,x)}}{\sqrt{c+d\,x}} = \frac{2}{d}\,\mathrm{Subst}\!\left[F^{g\,\left(e-\frac{c\,f}{d}\right)+\frac{f\,g\,x^2}{d}},\;x,\;\sqrt{c+d\,x}\right]\partial_x\sqrt{c+d\,x}$

Rule:

$$\int \frac{F^{g\,(e+f\,x)}}{\sqrt{c+d\,x}}\,dx\;\longrightarrow\;\frac{2}{d}\,\mathrm{Subst}\!\left[\int F^{g\,\left(e-\frac{c\,f}{d}\right)+\frac{f\,g\,x^2}{d}}\,dx,\;x,\;\sqrt{c+d\,x}\right]$$

Program code:

```
Int[F_^(g_.*(e_.+f_.*x_))/Sqrt[c_.+d_.*x_],x_Symbol] :=
  2/d*Subst[Int[F^(g*(e-c*f/d)+f*g*x^2/d),x],x,Sqrt[c+d*x]] /;
FreeQ[{F,c,d,e,f,g},x] && Not[TrueQ[$UseGamma]]
```

2: $\int (c+d\,x)^m\,F^{g\,(e+f\,x)}\,dx$ when $m\notin\mathbb{Z}$

Rule: If $2\,m\notin\mathbb{Z}$, then

$$\int (c+d\,x)^m\,F^{g\,(e+f\,x)}\,dx \longrightarrow -\frac{F^{g\left(e-\frac{c\,f}{d}\right)}\,(c+d\,x)^{\text{FracPart}[m]}}{d\left(-\frac{f\,g\,\text{Log}[F]}{d}\right)^{\text{IntPart}[m]+1}\left(-\frac{f\,g\,\text{Log}[F]\,(c+d\,x)}{d}\right)^{\text{FracPart}[m]}}\,\text{Gamma}\left[m+1,\,-\frac{f\,g\,\text{Log}[F]}{d}\,(c+d\,x)\right]$$

Program code:

```
Int[(c_.+d_.*x_)^m_*F_^(g_.*(e_.+f_.*x_)),x_Symbol] :=
  -F^(g*(e-c*f/d))*(c+d*x)^FracPart[m]/(d*(-f*g*Log[F]/d)^(IntPart[m]+1)*(-f*g*Log[F]*(c+d*x)/d)^FracPart[m])*
    Gamma[m+1,(-f*g*Log[F]/d)*(c+d*x)] /;
FreeQ[{F,c,d,e,f,g,m},x] && Not[IntegerQ[m]]
```

4: $\int (c+d\,x)^m\left(b\,F^{g\,(e+f\,x)}\right)^n dx$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\left(b\,F^{g\,(e+f\,x)}\right)^n}{F^{g\,n\,(e+f\,x)}} == 0$

Rule:

$$\int (c+d\,x)^m\left(b\,F^{g\,(e+f\,x)}\right)^n dx \longrightarrow \frac{\left(b\,F^{g\,(e+f\,x)}\right)^n}{F^{g\,n\,(e+f\,x)}}\int (c+d\,x)^m\,F^{g\,n\,(e+f\,x)}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(b_.*F_^(g_.*(e_.+f_.*x_)))^n_,x_Symbol] :=
  (b*F^(g*(e+f*x)))^n/F^(g*n*(e+f*x))*Int[(c+d*x)^m*F^(g*n*(e+f*x)),x] /;
FreeQ[{F,b,c,d,e,f,g,m,n},x]
```

2: $\int (c+d\,x)^m\left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$**, then**

$$\int (c+d\,x)^m\left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p dx \longrightarrow \int (c+d\,x)^m\,\text{ExpandIntegrand}\left[\left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p,\,x\right]dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(c+d*x)^m,(a+b*(F^(g*(e+f*x)))^n)^p,x],x] /;
FreeQ[{F,a,b,c,d,e,f,g,m,n},x] && IGtQ[p,0]
```

3: $\int \frac{(c+d\,x)^m}{a+b\,\left(F^{g\,(e+f\,x)}\right)^n}\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{a+b\,z}=\frac{1}{a}-\frac{b\,z}{a\,(a+b\,z)}$

Rule: If $m\in\mathbb{Z}^+$, **then**

$$\int \frac{(c+d\,x)^m}{a+b\,\left(F^{g\,(e+f\,x)}\right)^n}\,dx\,\longrightarrow\,\frac{(c+d\,x)^{m+1}}{a\,d\,(m+1)}-\frac{b}{a}\int \frac{(c+d\,x)^m\,\left(F^{g\,(e+f\,x)}\right)^n}{a+b\,\left(F^{g\,(e+f\,x)}\right)^n}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_./(a_+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.),x_Symbol] :=
  (c+d*x)^(m+1)/(a*d*(m+1)) - b/a*Int[(c+d*x)^m*(F^(g*(e+f*x)))^n/(a+b*(F^(g*(e+f*x)))^n),x] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && IGtQ[m,0]
```

x: $\int \frac{(c+d\,x)^m}{a+b\,\left(F^{g\,(e+f\,x)}\right)^n}\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Integration by parts

- **Basis:** $\frac{1}{a+b\,\left(F^{g\,(e+f\,x)}\right)^n}=-\partial_x\frac{\text{Log}\left[1+\frac{a}{b\,\left(F^{g\,(e+f\,x)}\right)^n}\right]}{a\,f\,g\,n\,\text{Log}[F]}$

Rule: If $m\in\mathbb{Z}^+$, **then**

$$\int \frac{(c+d\,x)^m}{a+b\,\left(F^{g\,(e+f\,x)}\right)^n}\,dx\,\longrightarrow\,-\frac{(c+d\,x)^m}{a\,f\,g\,n\,\text{Log}[F]}\,\text{Log}\left[1+\frac{a}{b\,\left(F^{g\,(e+f\,x)}\right)^n}\right]+\frac{d\,m}{a\,f\,g\,n\,\text{Log}[F]}\int (c+d\,x)^{m-1}\,\text{Log}\left[1+\frac{a}{b\,\left(F^{g\,(e+f\,x)}\right)^n}\right]\,dx$$

Program code:

```
(* Int[(c_.+d_.*x_)^m_./(a_+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.),x_Symbol] :=
  -(c+d*x)^m/(a*f*g*n*Log[F])*Log[1+a/(b*(F^(g*(e+f*x)))^n)] +
  d*m/(a*f*g*n*Log[F])*Int[(c+d*x)^(m-1)*Log[1+a/(b*(F^(g*(e+f*x)))^n)],x] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && IGtQ[m,0] *)
```

4: $\int (c + d x)^m \left(a + b \left(F^{g (e+f x)}\right)^n\right)^p dx$ when $p \in \mathbb{Z}^- \land m \in \mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis:** $(a + b z)^p = \frac{(a+b z)^{p+1}}{a} - \frac{b z (a+b z)^p}{a}$

- **Rule: If** $p \in \mathbb{Z}^- \land m \in \mathbb{Z}^+$**, then**

$$\int (c + d x)^m \left(a + b \left(F^{g (e+f x)}\right)^n\right)^p dx \longrightarrow$$
$$\frac{1}{a} \int (c + d x)^m \left(a + b \left(F^{g (e+f x)}\right)^n\right)^{p+1} dx - \frac{b}{a} \int (c + d x)^m \left(F^{g (e+f x)}\right)^n \left(a + b \left(F^{g (e+f x)}\right)^n\right)^p dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.)^p_,x_Symbol] :=
  1/a*Int[(c+d*x)^m*(a+b*(F^(g*(e+f*x)))^n)^(p+1),x] -
  b/a*Int[(c+d*x)^m*(F^(g*(e+f*x)))^n*(a+b*(F^(g*(e+f*x)))^n)^p,x] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && ILtQ[p,0] && IGtQ[m,0]
```

5: $\int (c + d x)^m \left(a + b \left(F^{g (e+f x)}\right)^n\right)^p dx$ when $m \in \mathbb{Z}^+ \land p < -1$

Derivation: Integration by parts

- **Rule: If** $m \in \mathbb{Z}^+ \land p < -1$**, let** $u = \int \left(a + b \left(F^{g (e+f x)}\right)^n\right)^p dx$**, then**

$$\int (c + d x)^m \left(a + b \left(F^{g (e+f x)}\right)^n\right)^p dx \longrightarrow u (c + d x)^m - d m \int u (c + d x)^{m-1} dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.)^p_,x_Symbol] :=
  With[{u=IntHide[(a+b*(F^(g*(e+f*x)))^n)^p,x]},
  Dist[(c+d*x)^m,u,x] - d*m*Int[(c+d*x)^(m-1)*u,x]] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && IGtQ[m,0] && LtQ[p,-1]
```

6. $\int u^m \left(a+b\left(F^{g\,v}\right)^n\right)^p dx$ when $v == e+f\,x \land u == (c+d\,x)^q$

1: $\int u^m \left(a+b\left(F^{g\,v}\right)^n\right)^p dx$ when $v == e+f\,x \land u == (c+d\,x)^q \land m \in \mathbb{Z}$

Derivation: Algebraic normalization

Rule: If $v == e+f\,x \land u == (c+d\,x)^q \land m \in \mathbb{Z}$**, then**

$$\int u^m \left(a+b\left(F^{g\,v}\right)^n\right)^p dx \longrightarrow \int (c+d\,x)^{m\,q} \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p dx$$

Program code:

```
Int[u_^m_.*(a_.+b_.*(F_^(g_.*v_))^n_.)^p_.,x_Symbol] :=
  Int[NormalizePowerOfLinear[u,x]^m*(a+b*(F^(g*ExpandToSum[v,x]))^n)^p,x] /;
FreeQ[{F,a,b,g,n,p},x] && LinearQ[v,x] && PowerOfLinearQ[u,x] && Not[LinearMatchQ[v,x] && PowerOfLinearMatchQ[u,x]] && IntegerQ[m]
```

2: $\int u^m \left(a+b\left(F^{g\,v}\right)^n\right)^p dx$ when $v == e+f\,x \land u == (c+d\,x)^q \land m \notin \mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{\left((c+d\,x)^q\right)^m}{(c+d\,x)^{m\,q}} == 0$
- **Rule: If** $v == e+f\,x \land u == (c+d\,x)^q \land m \notin \mathbb{Z}$**, then**

$$\int u^m \left(a+b\left(F^{g\,v}\right)^n\right)^p dx \longrightarrow \frac{\left((c+d\,x)^q\right)^m}{(c+d\,x)^{m\,q}} \int (c+d\,x)^{m\,q} \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p dx$$

- **Program code:**

```
Int[u_^m_.*(a_.+b_.*(F_^(g_.*v_))^n_.)^p_.,x_Symbol] :=
  Module[{uu=NormalizePowerOfLinear[u,x],z},
  z=If[PowerQ[uu] && FreeQ[uu[[2]],x], uu[[1]]^(m*uu[[2]]), uu^m];
  uu^m/z*Int[z*(a+b*(F^(g*ExpandToSum[v,x]))^n)^p,x]] /;
FreeQ[{F,a,b,g,m,n,p},x] && LinearQ[v,x] && PowerOfLinearQ[u,x] && Not[LinearMatchQ[v,x] && PowerOfLinearMatchQ[u,x]] &&
  Not[IntegerQ[m]]
```

X: $\int (c+d\,x)^m \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p\,dx$

- **Rule:**

$$\int (c+d\,x)^m \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p\,dx \;\longrightarrow\; \int (c+d\,x)^m \left(a+b\left(F^{g\,(e+f\,x)}\right)^n\right)^p\,dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^m_.*(a_+b_.*(F_^(g_.*(e_.+f_.*x_)))^n_.)^p_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*(a+b*(F^(g*(e+f*x)))^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x]
```